Wprowadzone w tym odcinku „słówka" możemy przetłumaczyć następująco:
proof: dowód
let *x* **be** *punkt*: niech *x* będzie *punktem*
such that: taki, że
assume (that): załóżmy (że)
and: i, oraz
thus: a zatem, więc
end: koniec (dowodu, rozumowania).

Przypominamy:
Każdy, kto nadeśle pod adresem redakcji rozwiązanie wraz z zaadresowaną do siebie kopertą — lepiej większą — z naklejonym znaczkiem, otrzyma wydruk z komputera z komentarzem do tego rozwiązania.

Mizar — MSE (3)

W poprzednim odcinku pokazaliśmy, jak zapisywać w Mizarze poszczególne kroki rozumowania. Kroki mogą się składać na większą całość, np. na dowód. Zasadniczą strukturę dowodów w Mizarze przedstawimy dzisiaj. Oczywiście najpierw musimy sformułować tezę naszego dowodu. Następnie piszemy **proof** i ... ano właśnie, korzystając z założonych aksjomatów bądź z tego, co wcześniej już stwierdziliśmy, dowodzimy naszej tezy. Cóż to znaczy? Jak to zrobić? Zależy to istotnie od struktury aktualnie dowodzonego zadania (tezy). Najprostsze przypadki to, gdy teza jest koniunkcją lub implikacją. Jeśli jest koniunkcją, to musimy w dowodzie stwierdzić prawdziwość każdego jej członu. (Wszystkie dzisiejsze przykłady zakładają wstęp z zadania z poprzedniego odcinka.)
Np:

```
NW[A,C] & NW[B,C]
 PROOF
  THUS NW[A,C] BY Z2;
  THUS NW[B,C] BY Z3
 END;
```

Dowód kończymy słowem **end**. Słowo **thus** umieszczone w dowodzie służy wskazaniu, że następujące po nim zdanie jest konkluzją — tym (lub częścią tego), co mieliśmy udowodnić. W powyższym przypadku mogliśmy napisać, **thus** NW[*a*, *c*] & NW [*b*, *c*] **by** *Z2*, *Z3* co też by wystarczyło.

Zilustrujmy teraz przypadek, gdy teza jest implikacją. Zakładamy wtedy, że prawdziwy jest jej poprzednik i staramy się teraz wywnioskować następnik. Na przykład:

```
NW[A,B] &NW[C,D] IMPLIES NW[A,B]
 PROOF
  ASSUME A: NW[A,B] & NW[C,D];
  THUS NW[A,B] BY A
 END;
```

Słówko **assume** wskazuje założenie w dowodzie. Słówko **thus** wskazuje na wywnioskowany następnik. Założenie może składać się z całej listy (być może ponumerowanych — etykietowanych) zdań, łączonych słówkiem **and**. Np:

```
NW[A,B] & (NW[D,C] OR NW[A,E]) IMPLIES NW[A,D] &  NW[D,C]
 PROOF
  ASSUME THAT A: NW[A,B]   AND
              B: NW[D,C] OR NW[A,E];
  THUS NW[A,D] BY T1;
  THUS NW[D,C] BY B,Z5
 END;
```

W powyższym przykładzie, z chwilą założenia (przez **assume** ...) poprzednika implikacji pozostało do udowodnienia zdanie będące koniunkcją dwu członów (tzn. następnik wyjściowej tezy). Dowód moglibyśmy zakończyć przez konkluzję w postaci dokładnie takiej, jak wygląda następnik tezy (tj. **thus** NW [*a*, *d*] & NW [*d*, *c*] **by** ... ?). Możemy również, i tak zrobiliśmy, zakończyć dwiema konkluzjami stwierdzającymi oddzielnie poszczególne człony następnika (czasami tak jest wygodnie).

Poprzednie przykłady ilustrowały użycie dowodów w przypadkach, kiedy to nie było niezbędne. Dowód jest koniecznie potrzebny przy dowodzie w Mizarze zdań z kwantyfikatorem ogólnym, choćby z tego powodu, że dla maszyny (checkera) żadne zdanie ogólne nie jest oczywiste i wymaga uzasadnienia — na ogół w postaci dowodu.
W codziennej praktyce, jeśli chcemy udowodnić zdanie ogólne, to najczęściej bierzemy pod uwagę dowolne obiekty odpowiednich typów; zakładamy o nich, że spełniają ograniczenia nałożone w kwantyfikatorze i z tymi aktywami usiłujemy udowodnić kwantyfikowane zdanie.

Na początek weźmy prosty przykład. Gdyby ze wstępu usunąć zwrotność, to można by i tak jej dowieść.

```
FOR X BEING ULAMEK HOLDS NW[X,X]
 PROOF
  LET X1 BE ULAMEK;
   A: NW[X1,X1] OR NW[X1,X1] BY SPOJNOSC;
  THUS NW[X1,X1] BY A
 END;
```

Wybraliśmy pewien ułamek *x1*, dowolny, lecz ustalony na czas reszty dowodu (**let** *x1* **be** *ułamek*). Następnie pokazaliśmy pewien pomocniczy fakt o *x1* — i fakt ten nie był ani założeniem, ani konkluzją w tym dowodzie. Po czym stwierdziliśmy o wybranym dowolnie *x1* to, co chcieliśmy wiedzieć o każdym ułamku. Poniższy przykład jest bardziej obszerny:

```
FOR X,Y,Z BEING ULAMEK
   ST NW[X,Y] & X<>Y & NW[Z,X]
     HOLDS NOT NW[Y,X] & NW[Z,Y]
 PROOF
  LET X,Y,Z BE ULAMEK SUCH THAT
   A: NW[X,Y]  AND  B: X<>Y  AND  C: NW[Z,X];
   D: NOT (NW[X,Y] & NW[Y,X]) BY B,ANTYSYMETRIA;
  THUS NOT NW[Y,X] BY D,A;
   E: NW[Z,X] & NW[X,Y] BY A,C;
  THUS  NW[Z,Y] BY E,PRZECHODNIOSC
 END;
```

Założenia odpowiadające ograniczeniu kwantyfikatora mogliśmy ująć w równoważny sposób tak:

```
LET X,Y,Z BE ULAMEK;
ASSUME THAT A: NW[X,Y]  AND  B: X<>Y  AND  C: NW[Z,X]
```

Konstrukcja rozpoczynająca się od **let** jest nazywana deklaracją.

Jak widzimy, dowód jest ciągiem założeń i deklaracji (nie podlegających uzasadnieniu) oraz stwierdzeń (wśród nich konkluzji). Kolejność występowania tych elementów jest narzucana przez postać dowodzonej tezy. Stwierdzenia są uzasadniane np. w sposób podany w poprzednim odcinku i używany w tym. Jak je uzasadniać inaczej, podamy później.

Zadania:
Przyjmując wstęp podany w drugim odcinku dla teorii liniowego porządku dla ułamków udowodnić:

```
T8: FOR X,Y,Z BEING ULAMEK
       ST NW[X,Y] & NW[Y,Z] & NW[Z,X] HOLDS X=Y    ;
T9: FOR X,Y,Z BEING ULAMEK
       ST NOT NW[X,Y] & NOT NW[Y,Z] HOLDS NW[Z,X]  ;
T10: FOR X,Y,Z BEING ULAMEK
        ST NOT NW[X,Z] & NW[Y,Z] HOLDS NOT NW[X,Y];
```

dr Krzysztof PRAŻMOWSKI, dr Piotr RUDNICKI